
ABSTRACT 
Propensity score matching (PSM) is commonly used in observational studies to reduce the potential selection bias. 
The process of PMS includes propensity score estimation, matching and evaluation. This paper presents a case 
application in outcome evaluation of medication therapy management at retail pharmacy. It illustrates how baseline 
outcome balance check could help detect potential omissions of covariates that could affect both the treatment and 
the outcomes. Further, it shows how in retail MTM setting, it is possible to find a common set of covariates that affect 
multiple studies outcomes and perform a single propensity score matching rather than a separate matching for each 
individual study outcome.  

INTRODUCTION:  
Propensity Score Matching is widely used in observational studies. The basic idea of PSM is to match an untreated 
group to the treated group such that the matched group is comparable to the treated group in all aspects of 
characteristics except the treatment. The observed outcome for the matched group can then be used as the 
counterfactual, and the treatment effect is estimated simply as the difference between the average outcomes of the 
two groups.  

The challenge of PSM is that the propensity score is unknown and must be estimated with a propensity score model. 
The quality of propensity score model is evaluated solely on the covariates balance between the treated and 
untreated groups. Conventional balance tests typically check for differences in average covariates between the 
treated and un-treated groups to see if there remain any significant differences between the two groups after 
propensity score matching. 

It is obviously important that the propensity score model balances the covariates. But how do we ensure that all 
covariates relevant to the study are incorporated into the model? The current paper attempts to address this very 
issue by way of a case application, in which we examined multiple baseline study outcomes to verify that the PSM 
model is correctly specified with all relevant covariates. Conventionally, when a study involves multiple outcomes, 
PSM is performed for each individual outcome. In our current case application, a single PSM was done for multiple 
study outcomes. We then check for balance in all study outcomes at baseline, between the treated and un-treated 
groups, after propensity score matching. A correctly specified PSM model should balance all study outcomes at 
baseline.  

CASE APPLICATION IN MEDICATION THERAPY MANAGEMENT AT RETAIL PHARMACY 
A medication therapy management program at a national retail pharmacy chain provides its patients comprehensive 
medication review (CMR) to promote safe and effective use of medications. During a review session, a store 
pharmacist meets one-on-one with his patient to: 1.organize the patient’s medications, 2.evaluate the patient’s 
medications for cost-saving alternatives, 3.review medications for side effects or harmful drug-drug interactions, 
4.provide patient with a list of active medications the patient is on, 5. educate the patient on the importance of 
medication adherence, answer any questions that patient may have about his prescriptions, and coordinate any 
medication changes with his primary physician. 

A retrospective cohort study with a propensity matched control was conducted to evaluate the effects of CMR on 
brand to generic drug savings, medication adherence and immunization rates among the study subjects. The 
treatment group consisted of patients who had at least one completed CMR between Jan 2011 and Jun 2011.  The 
control group consisted of patients who received no CMR and had at least one prescription filled at Walgreens 
between Jan 2011 and Jun 2011.  A 3.5% random sample (1,482,330 patients) was drawn from Walgreens EDW to 
serve as the pool for control group. 

BASELINE COVARIATES BEFORE PROPENSITY SCORE MATCHING  
As numbers shown in Table 1, the baseline covariates were not balanced between the treatment and control groups 
before propensity matching.  

Table 1. Baseline Covariates Before Propensity Score Matching 

Covariates 
Treatment (7,229) Control (1,482,330) p-value 

N % N %   

Age Under 18 4 0.1% 356,453 24.0%   

  [18, 36) 58 0.8% 356,264 24.0%   

  [36, 50) 297 4.1% 288,926 19.5%   

  [50, 65) 1,219 16.9% 284,825 19.2%   

  [66, 80) 4,308 59.6% 145,887 9.8%   

  [80+  1,343 18.6% 49,975 3.4% < 0.0001 

  Mean Age (S.D.) 70.7 11.0 37.7 23.0   

Gender Female 4,437 61.4% 848,814 57.3%   

  Male 2,792 38.6% 633,516 42.7% < 0.0001 

Insurance Plan Cash        12 0.2% 157,901 10.7%   

  Commercial  1,960 27.1% 662,915 44.7%   

  Medicaid    7 0.1% 186,674 12.6%   

  Medicare    5,243 72.5% 406,496 27.4%   

  Other       0 0.0% 5,142 0.3%   

  PSC         4 0.1% 32,665 2.2%   

  Tricare     3 0.0% 30,537 2.1% < 0.0001 

# of Therapeutics <= 3  680 9.4% 1,075,348 72.5%   

  <= 6  1,407 19.5% 264,230 17.8%   

  <= 9  1,921 26.6% 90,943 6.1%   

  >= 10 3,221 44.6% 51,809 3.5% < 0.0001 

Copay Mean (S.D.) $384  $489  $84  $221  < 0.0001 

Total Drug Costs Mean (S.D.) $2,181  $2,151  $398  $1,062  < 0.0001 

# Competing Business  Mean (S.D.) 4.54 3.46 4.59 3.27 0.29 

Community Income Under Class     10 0.1% 1,121 0.1%   

  Working poor     171 2.4% 19,333 1.3%   

  Working class     4,081 56.5% 699,476 47.2%   

  Lower middle class 2,919 40.4% 733,112 49.5%   

  Upper middle class  48 0.7% 29,288 2.0% < 0.0001 

Community Race Mix % White    0.79 0.2 0.79 0.2 0.14 

  % Asian    0.03 0.06 0.04 0.07 < 0.0001 

  % Black    0.15 0.19 0.14 0.2 < 0.0001 

  % Hispanic 0.17 0.22 0.18 0.22 < 0.0001 

BASELINE COVARIATES AFTER PROPENSITY SCORE MATCHING  
The initial PSM model includes the following baseline covariates  

 Age 

 Gender 

 Co-pay 

 Total Drug Cost 

 Average Cost per Rx 

 Number of therapeutic classes  

 Insurance Plan 

 Average community Income  

 Community Race composition  

 

Different methods can be used to estimate propensity score: probit / logistics models, classification trees, neural 
networks etc. We opted for SAS PROC LOGISTIC and below are the SAS codes for the initial PSM model with 
the above listed covariates and the treatment assignment variable EVENT. The estimated propensity scores are 
stored in the variable PROB and saved in the SAS dataset PSMScore 

 
%let predictors = Agecat gndr_cd copay total_drug_cost RevenuePerRx nTherapClass Insurance  
                  incomegroup __White __Asian __Black __Hispanic; 
 
proc logistic data = &PSMdata descending; 
     class   Agecat gndr_cd nTherapClass Insurance incomegroup       
             nTherapClass; 
     model   Event=&predictors; 
     output  out = PSMScore(drop=_LEVEL_ index=(event=(event))) prob=prob; 
 run;  

 
We used caliper matching algorithm to match the control group to the treatment group on a 1-to-1 ratio without 
replacement. The algorithm adjusts the caliper width from 4 to 1 decimal places so that it finds the "best" match 
first and the "next-best" match next. In the event of multiple matches, the algorithm randomly selects a control 
for the treatment.  

The actual SAS codes are included in the appendix with the following parameters: 

 InData: input SAS data set for propensity score matching 

 OutData: output SAS data set after propensity score matching 

 DependentVar: dependent variable for group assignment (0,1 coded, 1=treatment 0=control)  

 EstimatedProb : variable holding propensity score value 

 N: propensity matching ratio (1:N matching)   

 MaxDigit : starting decimal places used to match propensity score 

 PickupMethod : either select the first (FIXED) or a random control for the treatment In the event of multiple controls 

 Method:  matching with or without replacement (REPLACEMENT or NOREPLACEMENT) 

Covariates balance check after propensity score matching indicated that the baseline covariates were 
comparable between the two groups (Table 2). 

 
Table 2. Baseline Covariates After Propensity Score Matching 

Covariates 
Treatment (7,166) Control (7,166) p-value 

N % N %   

Age Under 18 4 0.1% 7 0.1%   

  [18, 36) 58 0.8% 60 0.8%   

  [36, 50) 297 4.1% 287 4.0%   

  [50, 65) 1,218 17.0% 1,258 17.6%   

  [66, 80) 4,253 59.3% 4,265 59.5%   

  [80+  1,336 18.6% 1,289 18.0% 0.77 

  Mean Age (S.D.) 70.7 11.1 70.2 11.6   

Gender Female 4,395 61.3% 4,379 61.1%   

  Male 2,771 38.7% 2,787 38.9% 0.78 

Insurance Plan Cash        12 0.2% 5 0.1%   

  Commercial  1,956 27.3% 2,000 27.9%   

  Medicaid    7 0.1% 3 0.0%   

  Medicare    5,184 72.3% 5,148 71.8%   

  Other       0 0.0% 4 0.1%   

  PSC         4 0.1% 3 0.0%   

  Tricare     3 0.0% 3 0.0% 0.16 

# of Therapeutics <= 3  680 9.5% 631 8.8%   

  <= 6  1,406 19.6% 1,393 19.4%   

  <= 9  1,910 26.7% 1,903 26.6%   

  >= 10 3,170 44.2% 3,239 45.2% 0.45 

Copay Mean (S.D.) $383  $487  $396  $811  0.23 

Total Drug Cost Mean (S.D.) $2,165  $2,093  $2,243  $3,670  0.12 

# Competitors  Mean (S.D.) 4.54 3.47 4.53 3.35 0.90 

Community Income Under Class     10 0.1% 12 0.2%   

  Working poor     171 2.4% 172 2.4%   

  Working class     4,040 56.4% 4,066 56.7%   

  Lower middle class 2,897 40.4% 2,855 39.8%   

  Upper middle class  48 0.7% 61 0.9% 0.71 

Community Race Mix % White Mean (S.D.)   0.79 0.20 0.79 0.20 0.70 

  % Asian Mean (S.D.)   0.03 0.06 0.03 0.06 0.29 

  % Black Mean (S.D.)   0.15 0.19 0.15 0.19 0.37 

  % Hispanic Mean (S.D.) 0.17 0.23 0.17 0.24 0.20 

*Chi‐squared test and paired t test were used  

BASELINE OUTCOMES CHECK:   
However, baseline study outcomes check revealed that even though the propensity score model balanced the 
baseline covariates, it did not balance the baseline outcomes generic dispense rates (GDR). As was shown in 
table 3, the baseline GDR for the treatment group was nearly twice that of the control group after propensity 
score matching.   

Table3: Baseline Generic Dispense Rate after Propensity Score Matching 

Group # Patients Baseline Generic Dispense Rate P-value 

Treatment 1367 11.7%  

Control 1383 6.7% <.0001 

 

We could incorporate baseline GDR into the propensity score model to force a baseline GDR balance. However, to 
do so would most likely require us to do a separate propensity matching for each study outcomes. In addition, 
baseline GDR was balanced does not necessarily mean the model is correctly specified with all relevant covariates. 
It’s important to try to uncover any additional confounders that was not considered in the model and could potentially 
affect the study outcomes. 

BASELINE COVARIATES RECHECK AFTER PROPENSITY SCORE MODEL ADJUSTMENT  
A closer examination of the data revealed that the index dates, which defined GDR baselines for the two groups, 
were not balanced. We then added the index month in the propensity model, re-matched the two groups and checked 
the baseline GDR again. As was shown in table 4, the baseline GDR now appeared to be balanced.  

Table 3. Baseline Covariates After Propensity Score adjustment 
 

Covariates 
Treatment (7,142) Control (7,142) p-value 

N % N %   

Age Under 18 4 0.1% 10 0.1%   

  [18, 36) 58 0.8% 54 0.8%   

  [36, 50) 297 4.2% 290 4.1%   

  [50, 65) 1,219 17.1% 1,226 17.2%   

  [66, 80) 4,230 59.2% 4,238 59.3%   

  [80+  1,334 18.7% 1,324 18.5% 0.72 

  Mean Age (S.D.) 70.6 11.1 70.3 11.7 0.08 

Gender Female 4,381 61.3% 4,394 61.5%   

  Male 2,761 38.7% 2,748 38.5% 0.82 

Insurance Plan Cash        12 0.2% 13 0.2%   

  Commercial  1,956 27.4% 1,973 27.6%   

  Medicaid    7 0.1% 7 0.1%   

  Medicare    5,160 72.2% 5,140 72.0%   

  Other       0 0.0% 6 0.1%   

  PSC         4 0.1% 2 0.0%   

  Tricare     3 0.0% 1 0.0% 0.25 

# of Therapeutics <= 3  680 9.5% 635 8.9%   

  <= 6  1,407 19.7% 1,438 20.1%   

  <= 9  1,907 26.7% 1,941 27.2%   

  >= 10 3,148 44.1% 3,128 43.8% 0.52 

MTM Complete Month 01/2011 428 6.0% 412 5.8%   

  02/2011 727 10.2% 670 9.4%   

  03/2011 1,165 16.3% 1,209 16.9%   

  04/2011 1,531 21.4% 1,552 21.7%   

  05/2011 1,575 22.1% 1,600 22.4%   

  06/2011 1,716 24.0% 1,699 23.8% 0.57 

Copay Mean (S.D.) $384  $488  $386  $704  0.85 

Total Drug Cost Mean (S.D.) $2,170  $2,134  $2,103  $3,445  0.16 

Revenue Per Rx Mean (S.D.) $70  $219  $68  $151  0.56 

Neighborhood Income Under Class     10 0.1% 9 0.1%   

  Working poor     170 2.4% 171 2.4%   

  Working class     4,029 56.4% 4,081 57.1%   

  Lower middle class 2,885 40.4% 2,832 39.7%   

  Upper middle class  48 0.7% 49 0.7% 0.93 

Neighborhood Race % White Mean (S.D.)   0.79 0.20 0.79 0.20 0.60 

  % Asian Mean (S.D.)   0.03 0.06 0.03 0.05 0.99 

  % Black Mean (S.D.)   0.15 0.19 0.15 0.20 0.66 

  % Hispanic Mean (S.D.) 0.17 0.23 0.17 0.23 0.70 

*Chi‐squared test and paired t test were used  

 

Table 4: Baseline Generic Dispense Rate after Final Propensity Score Matching 

Group # Patients Baseline Generic Dispense Rate P-value 

Treatment 1501 11.7%  

Control 1481 10.9% 0.36 

 

OTHER BASELINE OUTCOMES:   
We then checked the baseline pneumovax and zoster immunization rates (table 5 and 6) and again they appeared to 
be balanced between the two groups.  

 

Table 5 Baseline Pneumovax immunization Rate after Final Propensity Score Matching 

 

Group #  Patients Baseline Pneumovax immunization Rate  P-value 

Treatment 7,142 0.3%  

Control 7,142 0.3% 0.75 

Table 6 Baseline Zoster immunization Rate after Final Propensity Score Matching 

 

Group #  Patients Baseline Zoster immunization Rate P-value 

Treatment 7,142 0.4%  

Control 7,142 0.4% 1.00 

 

Lastly, we checked the baseline medication adherence as measured in proportion of days covered or PDC for 5 
select drug categories (table 7) and once more they appeared to be balanced between the two groups. 

Table 7 Baseline Proportion of Days Covered (PDC) after Final Propensity Score Matching 

Drug Category 
Treatment Group Control Group 

p-value 
N PDC N PDC 

ANTIDIABETICS 3,320 0.87 2,008 0.87 0.87 

ANTIHYPERLIPIDEMICS 3,792 0.89 2,745 0.90 0.24 

ANTIHYPERTENSIVES 3,693 0.91 2,814 0.91 0.82 

BETA BLOCKERS 2,205 0.91 1,782 0.90 0.19 

CALCIUM CHANNEL BLOCKERS 1,500 0.92 1,137 0.92 1.00 

OVERALL 14,510 0.90 10,486 0.90 0.30 

  

CONCLUSION 
The propensity score model balanced all the study outcomes on top of the covariates, which, to a degree, validated 
the model and we were able to conduct the study as if it were a pseudo randomized study with just a single matching 
to evaluate multiple outcomes. 

Appendix 

 

%macro _MPSMSingleMatch(Out =, InCase= , InControl=, EstimatedProb = prob, No=No, Recordno= recordno, Digit = 5) ; 

          * Step 0: data  preparation; 

          data _null_; set &InCase; 

if _n_=&recordno then do; 

        call symput('_LocalPSScore' ,roundP); call symput('M_CaseSeqn'    , seqn); call symput('M_CaseUniqueID', UniqueID); 
 end; 

           run; 

         * Step 1: in the event of multiple match s, first match is selected; 

         %let M_Status=fail; %local iOrder; %let iOrder = 1; %let total  = 0; 

     %if %upcase(&PickupMethod) ne FIXED %then %do; 

      %let total = 0; 

          proc sql noprint; 

                select count(*) into :total from &InControl where roundP = &_LocalPSScore; 

          quit;   

          data _null_; if &total >= 1 then do; call symput('iOrder', floor(1 + &total*ranuni(9646429)));  end;  run; 

    %end; 

    %put total = &total iOrder = &iOrder; 

    data tmp(keep=CaseUniqueID MatchID CaseSeqn ControlSeqn);                    

            set &InControl(where=( roundP=&_LocalPSScore )) nobs=total; 

            if total>=1 then call symput('M_Status',"succeed");  

            if  _n_  =&iOrder then do; 

   CaseUniqueID=&M_CaseUniqueID; 

   MatchID     = &Digit; 

   CaseSeqn    = &M_CaseSeqn; 

                    ControlSeqn = seqn;   

                    call symput('M_ControlSeqn',seqn); 

                    output;  

            end; 

     run; 

    * Step 3: Remove selected observations from the control pool; 

    %if &M_Status eq succeed %then %do; 

        %if %upcase(&method) eq NOREPLACEMENT %then %do; 

 data &InControl; 

         set &InControl; 

         if seqn = &M_ControlSeqn then delete; 

 run; 

         %end; 

         proc append base=&Out data=tmp force;run;  

         proc sql; 

 update &InCase  

                  set status = 'YES' 

                  where seqn =  &M_CaseSeqn;  

          quit; 

%end; 

%mend _MPSMSingleMatch; 

 

%macro _mPSMatch(Indata=, OutData=, DependentVar=, EstimatedProb=prob, N=1, MaxDigit= 6, PickupMethod=Fixed 

               , method=NOREPLACEMENT); 

            OPTION SPOOL nosymbolgen nomprint nomlogic; 

            * Step 0: Generate case and Control data set with a unique ID; 

           data _case(drop=EventCount NonEventCount)  _control(drop=EventCount NonEventCount); 

  set   &Indata; 

  where &EstimatedProb ne . ; 

 retain EventCount    1; retain NonEventCount 1; 

 if    &DependentVar =1 then do; UniqueID = EventCount; output _case; EventCount = EventCount + 1; end; 

                  else  do; UniqueID = NonEventCount; output _control; NonEventCount = NonEventCount + 1; end; 

           run; 

           %global _nTotalCases _nMatchedCases; 

 proc sql noprint;select count(*) into :_nTotalCases from _Case; quit; 

                  %local k; 

 %do k=1 %to &N; 

         proc sort data=_case       ;  by &EstimatedProb; run; 

         proc sort data=_control   ;  by &EstimatedProb; run; 

         data _case    _CaseOrig   ;  set _case   ; seqn = _n_;Status = 'NO ';run; 

         data _control _ControlOrig;  set _control; seqn = _n_;Status = 'NO ';run; 

         * Step 1: make sure the output file is new; 

         %if %sysfunc(exist(_TMPMatchedOut)) %then %do; 

                  proc sql; drop table _TMPMatchedOut; quit; 

          %end;              

          * Step 2: Matching with varied Caliper width / decimal places; 

          %local i j; 

          %do i=1 %to &MaxDigit; 

  %let Number =0; 

  proc sql noprint;select count(*) into :Number from _case;quit; 

  data _case; set _case; roundP = round(&EstimatedProb,10**(- %eval(&MaxDigit - &i + 1) )); run; 

  data _control; set _control; roundP = round(&EstimatedProb,10**(- %eval(&MaxDigit - &i + 1) )); run; 

  %do j=1 %to &Number; 

          %_MPSMSingleMatch(Out =_TMPMatchedOut, InCase=_case, InControl=_control 

                                                                                  , EstimatedProb = prob, Recordno= &j, Digit = %eval(&MaxDigit - &i + 1) ); 

  %end; 

  data _case; set _case; where status ^= 'YES'; run; 

           %end; 

           * Merge Original Data together; 

           data Link; set _TMPMatchedOut; PairID = _n_; run; 

           proc sql; 

  create table _case&k(drop=seqn status) as 

            select _caseOrig.*   ,Link.MatchID,Link.CaseUniqueID,Link.PairID  from _caseOrig   ,Link  

                                              where  _caseOrig.seqn    = Link.CaseSeqn; 

  create table _control&k(drop=seqn status) as 

             select _ControlOrig.*,Link.MatchID,Link.CaseUniqueID,Link.PairID  

                                               from _ControlOrig,Link where  _ControlOrig.seqn = Link.ControlSeqn; 

                  create table _case  as  select * from _caseOrig; 

  create table _control as 

              select _ControlOrig.* from _ControlOrig where  _ControlOrig.seqn not in (select ControlSeqn from Link); 

                            quit; 

           * Prepare output; 

            %if &k eq 1 %then %do; 

                                   data _CaseOut;  set _case&k;  run; 

                                   data _ControlOut; set _control&k; run; 

            %end; 

            %else %do; 

  proc sql; 

          create table _CaseOut as 

          select _caseout.* from _CaseOut as A   , _case&k as B  where A.CaseUniqueID=B.CaseUniqueID; 

                                            create table _ControlOut as  

                                             select A.* from _ControlOut as A, _control&k as B where A.CaseUniqueID=B.CaseUniqueID 

                                             union 

           select B.* from _ControlOut as A, _control&k as B where A.CaseUniqueID=B.CaseUniqueID; 

  quit;   

               %end; 

                         %end;   

         data &outData(drop=UniqueID PairID rename=(MatchID=MatchPrecision CaseUniqueID=GroupID)  

                                                     index=(GroupID=(GroupID) &DependentVar=(&DependentVar))); 

  set _CaseOut _ControlOut; 

         run; 

         proc sql noprint;select count(*) into :_nMatchedCases from _CaseOut; quit; 

                          *  Remove temp files from the matching process; 

         proc sql noprint; 

                 drop table _case,_control,Link,Tmp,_Caseorig,_Controlorig,_tmpmatchedout; 

         quit; 

%mend  _mPSMatch; 

 

%_mPSMatch(InData=PSMScore, OutData= data.CMRMatch, DependentVar=event, EstimatedProb = prob, N = 1, MaxDigit = 4,  

                         PickupMethod = RANDOM, method = NOREPLACEMENT); 

